## CSCI 2021: x86-64 Control Flow

Chris Kauffman

Last Updated: Wed Mar 2 10:38:50 PM CST 2022

## Logistics

### Reading Bryant/O'Hallaron

- ► Ch 3.6: Control Flow
- ► Ch 3.7: Procedure calls

### Goals

- Finish Assembly Basics
- Jumps and Control flow
- Comparison / Test Instructions
- Procedure calls
- Stack Manipulation

### Assignments

- Lab07 asm coding
- ► HW07 asm debugging

## Project 2: Due $3/1 \frac{2}{28}$

- Problem 1: Thermometer C Functions (50%)
- Problem 2: Puzzlebox via GDB (50%)

### Project 3 later this week

## Control Flow in Assembly and the Instruction Pointer

- ▶ No high-level conditional or looping constructs in assembly
- Only %rip: Instruction Pointer or "Program Counter": memory address of the next instruction to execute
- Don't mess with %rip by hand: automatically increases as instructions execute so the next valid instruction is referenced
- Jump instructions modify %rip to go elsewhere
- Typically label assembly code with positions of instructions that will be the target of jumps
- Unconditional Jump Instructions always jump to a new location.
- Comparison / Test Instruction, sets EFLAGS bits indicating relation between registers/values
- ► Conditional Jump Instruction, jumps to a new location if certain bits of EFLAGS are set, ignored if bits not set

# Exercise: Loop Sum with Instruction Pointer (rip)

- Can see direct effects on rip in disassembled code
- rip increases corresponding to instruction length
- Jumps include address for next rip

61c: c3

```
// C Code equivalent
int sum=0, i=1, lim=100;
while(i<=lim){
   sum += i;
   i++;
}
return sum;</pre>
```

# rip 61c -> return address

```
00000000000005fa <main>:
ADDR. HEX-OPCODES
                           ASSEMBLY
                                                EFFECT ON RIP
5fa: 48 c7 c0 00 00 00 00
                           mov
                                  $0x0,%rax
                                              # rip = 5fa -> 601
601: 48 c7 c1 01 00 00 00
                                  $0x1,%rcx
                                              # rip = 601 -> 608
                           mov
608: 48 c7 c2 64 00 00 00
                                  $0x64,%rdx
                                              # rip = 608 -> 60f
                           mov
000000000000060f <LOOP>:
60f: 48 39 d1
                                  %rdx,%rcx
                                              # rip = 60f -> 612
                           cmp
612: 7f 08
                                  61c <END>
                                              \# rip = 612 -> 614 OR 61c
                           jg
                                              # rip = 614 -> 617
614: 48 01 c8
                                  %rcx,%rax
                           add
617: 48 ff c1
                           inc
                                  %rcx
                                              # rip = 617 -> 61a
61a: eb f3
                                  60f <LOOP>
                                              # rip = 61a -> 60f
                           jmp
000000000000061c <END>:
```

retq

## Disassembling Binaries

- Binaries hard to read on their own
- ▶ Many tools exist to work with them, notably objdump on Unix
- ► Can disassemble binary: show "readable" version of contents

```
> gcc -Og loop.s
                              # COMPILE AND ASSEMBLE
> file a.out
a.out: ELF 64-bit LSB pie executable, x86-64, version 1 (SYSV),
> objdump -d a.out
                              # DISASSEMBLE BINARY
        file format elf64-x86-64
a.out.:
Disassembly of section .text:
0000000000001119 <main>:
   1119:
               48 c7 c0 00 00 00 00
                                             $0x0.%rax
                                      mov
   1120: 48 c7 c1 01 00 00 00
                                             $0x1,%rcx
                                      mov
                                             $0x64, %rdx
   1127:
         48 c7 c2 64 00 00 00
                                      mov
000000000000112e <I.NOP>:
          48 39 d1
   112e:
                                             %rdx,%rcx
                                      cmp
            7f 08
   1131:
                                      ig
                                             113b <END>
         48 01 c8
   1133:
                                      add
                                             %rcx.%rax
   1136:
         48 ff c1
                                      inc
                                             %rcx
   1139:
               eb f3
                                             112e <LOOP>
                                      qmj
000000000000113b <END>:
   113b:
               c3
                                      retq
```

## FLAGS: Condition Codes Register

- Most CPUs have a special register with "flags" for various conditions
- ▶ In x86-64 this register goes by the following names

Name	Width	Notes
FLAGS	16-bit	Most important bits in first 16
EFLAGS	32-bit	Name shown in gdb
RFLAGS	64-bit	Not used normally

- Bits in FLAGS register are automatically set based on results of other operations
- Pertinent examples with conditional execution

Bit	Abbrev	Name	Description	
0	CF	Carry flag	Set if last op caused unsigned overflow	
6	ZF	Zero flag	Set if last op yielded a 0 result	
7	SF	Sign flag	Set if last op yielded a negative	
8	TF	Trap flag	Used by gdb to stop after one ASM instruction	
9	IF	Interrupt flag	1 7 0	
11	OF	Overflow flag		

## Comparisons and Tests

### Set the EFLAGS register by using comparison instructions

Name	Instruction	Examples	Notes
Compare	cmpX B, A	cmpl \$1,%eax	Like if(eax > 1){}
	Like: A - B	cmpq %rsi,%rdi	Like if(rdi > rsi){}
Test	testX B, A	testq %rcx,%rdx	Like if(rdx & rcx){}
	Like: A & B	testl %rax,%rax	Like if(rax){}

- Immediates like \$2 must be the first argument B
- ▶ B,A are NOT altered with cmp/test instructions
- ► EFLAGS register IS changed by cmp/test to indicate less than, greater than, 0, etc.

## Jump Instruction Summary

All control structures implemented using combination of Compare/Test + Jump instructions.

Instruction		
jmp LAB	Unconditional jump	-
je LAB	Equal / zero	ZF
jz LAB		ZF
jne LAB	Not equal $/$ non-zero	! ZF
jnz LAB		! ZF
js LAB	Negative ("signed")	SF
jns LAB	Nonnegative	!SF
jg LAB	Greater-than signed	!SF & !ZF
jge LAB	Greater-than-equal signed	!SF
jl LAB	Less-than signed	SF & !ZF
jle LAB	LAB Less-than-equal signed	
ja LAB	Above unsigned	!CF & !ZF
jae LAB	Above-equal unsigned	!CF
jb LAB	Below unsigned	CF & !ZF
jbe LAB	be LAB Below-equal unsigned	
jmp *OPER	jmp *OPER Unconditional jump to	
	variable address	

## **Examine:** Compiler Comparison Inversion

- Often compiler inverts comparisons
- i < n becomes cmpX /
  jge (jump greater/equal)</pre>
- i == 0 becomes cmpX /
  jne (jump not equal)
- ► This allows "true" case to fall through immediately
- Depending on structure, may have additional jumps
  - if(){ .. } usually has a single jump
  - if(){} else {} may have a couple

```
## Assembly translation of
## if(rbx >= 2){
## rdx = 10:
## }
## else{
## rdx = 5:
## }
## return rdx;
  cmpq $2,%rbx
                   # compare: rbx-2
        .LESSTHAN
                   # goto less than
 ## if(rbx \geq 2){
 movq $10, %rdx # greater/equal
 ## }
  qmj
        . AFTER
.LESSTHAN:
 ## else{
 movq $5,%rdx # less than
 ## }
. AFTER:
 ## rdx is 10 if rbx \geq= 2
 ## rdx is 5 otherwise
 movq %rdx,%rax
 ret.
```

### Exercise: The test Instruction

```
main:
                     $0.%eax
 2
            movl
 3
            Tvom
                     $5.%edi
 4
            movl
                     $3,%esi
                    $0.%rdx
 5
            mova
            movl
                     $-4,%ecx
            test1
                     %edi.%edi
 8
 9
            jnz
                      . NONZERO
            1bbs
                      $20.%eax
10
11
12
   . NONZERO:
            test1
                     %esi.%esi
13
                      FALSEY
14
            iz
15
            add1
                      $30, %eax
16
17
   .FALSEY:
18
            testq
                     %rdx,%rdx
19
            ie
                      .ISNULL
20
            add1
                      $40, %eax
21
    . TSNULL:
22
23
            testl
                     %ecx,%ecx
                      NONNEGATIVE
24
            ins
                      $50, %eax
25
            add1
26
27
    . NONNEGATIVE:
```

ret

28

- test1 %eax,%eax uses bitwise AND to examine a register
- Selected by compiler to check for zero, NULL, negativity, etc.
- Followed by je / jz / jne /
  jnz / js / jns
- Demoed in jmp\_tests\_asm.s
- Trace the execution
- Determine final value in %eax

### **Answers**: The test Instruction

```
1 ### From jmp_tests_asm_commented.s
 2 main:
 3
            movl
                    $0, %eax
                                     # eav is 0
                    $5,%edi
 4
            movl
                                    # set initial vals
                   $3,%esi
 5
            movl
                                     # for registers to
            movl
                   $0.%edx
                                     # use in tests
 6
                    $-4.%ecx
 7
            movl
 8
            ## eax=0, edi=5, esi=3, edx=NULL, ecx=-4
 9
10
            testl
                    %edi.%edi
                                     # anv bits set?
                     .NONZERO
11
            jnz
                                     # jump on !ZF (zero flag), same as jne
12
            ## if(edi == 0){
                    $20, %eax
13
            addl
14
            ## }
    . NONZERO:
15
                    %esi.%esi
                                     # anv bits set?
16
            testl
17
            iz
                     .FALSEY
                                     # jump on ZF same as je
            ## if(esi){
18
            addl
                    $30, %eax
19
            ## }
20
    .FALSEY:
21
22
            testq
                    %rdx,%rdx
                                     # any bits set
                     .ISNULL
                                     # same as iz: jump on ZF
23
            ie
24
            ## if(rdx != NULL){
25
            addl
                    $40,%eax
            ## }
26
    . TSNULL:
            testl
                    %ecx,%ecx
                                     # sign flag set on test to indicate negative results
28
                     .NONNEGATIVE
                                     # jump on !SF (not signed; e.g. positive)
29
            ins
            ## if(ecx < 0){
30
            add1
                    $50, %eax
31
32
            ## }
33
    NONNEGATIVE:
                          ## eax is return value
34
            ret
```

# cmov Family: Conditional Moves

- ► A family of instructions allows conditional movement of data into registers
- ► Can limit jumping in simple assignments

```
cmpq %r8,%r9
cmovge %r11,%r10 # if(r9 >= r8) { r10 = r11 }
cmovg %r13,%r12 # if(r9 > r8) { r12 = r13 }
```

- Note that condition flags are set on arithmetic operations
- cmpX is like subQ: both set FLAG bits the same
- Greater than is based on the SIGN flag indicating subtraction would be negative allowing the following:

```
subq %r8,%r9 # r9 = r9 - r8
cmovge %r11,%r10 # if(r9 >= 0) { r10 = r11 }
cmovg %r13,%r12 # if(r9 > 0) { r12 = r13 }
```

### Procedure Calls

#### Have seen basics so far:

```
main:
    call my_func # call a function
    ## arguments in %rdi, %rsi, %rdx, etc.
    ## control jumps to my_func, returns here when done
    ...

my_func:
    ## arguments in %rdi, %rsi, %rdx, etc.
    ...
    movl $0,%eax # set up return value
    ret # return from function
    ## return value in %rax
    ## returns control to wherever it came from
```

#### Need several additional notions

- Control Transfer to called function?
- Return back to calling function?
- Stack alignment and conventions
- Register conventions

## Procedure Calls Return to Arbitrary Locations

- call instructions always transfer control to start of return\_seven at line 4/5, like jmp instruction which modifies %rip
- ret instruction at line 6 must transfer control to different locations
  - 1. call-ed at line 11 ret to line 12
  - 2. call-ed at line 17 ret to line 18

ret cannot be a normal jmp

➤ To enable return to multiple places, record a Return Address when call-ing, use it when ret-urning

```
1 ### return seven asm.s
 2 text
3 .global return seven
  return_seven:
                  $7. %eax
         movl
                  ## jump to line 12 OR 18??
         ret
   .global main
8 main:
                  $8, %rsp
         subq
10
11
         call
                  return seven ## to line 5
         leag
                  .FORMAT 1(%rip), %rdi
12
                  %eax. %esi
13
         movl
                  $0, %eax
         movl
14
         call.
                  printf@PLT
15
16
         call
                  return seven ## to line 5
17
         lead
                  .FORMAT_2(%rip), %rdi
18
                  %eax, %esi
         movl
                  $0, %eax
20
         movl
21
         call
                  printf@PLT
22
                  $8, %rsp
23
         addq
                  $0. %eax
24
         movl
25
         ret
26 .data
27 .FORMAT 1: .asciz "first:
28 .FORMAT 2: .asciz "second: %d\n"
```

### call / ret with Return Address in Stack

#### call Instruction

- Push the "caller" Return Address onto the stack Return address is for instruction after call
- Change rip to first instruction of the "callee" function

#### ret Instruction

- Set rip to Return Address at top of stack
- 2. Pop the Return Address off the stack shrinking stack




Figure: Bryant/O'Hallaron Fig 3.26 demonstrates call/return in assembly

# return\_seven\_asm.s 1/2: Control Transfer with call

```
### BEFORE CALL
return seven:
   0x555555555139 <return seven> mov
                                          $0x7.%eax
   0x55555555513e <return seven+5> retq
main: ...
   0x555555555513f <main>
                                          $0x8, %rsp
                                   sub
=> 0x55555555555143  <main+4>
                                   calla
                                          0x5555555555139 <return_seven>
                                          0x2ee1(%rip),%rdi
   0x5555555555148 <main+9>
                                   lea
   0x555555555514f <main+16>
                                   mov
                                          %eax.%esi
(gdb) stepi
rsp = 0x7ffffffffe450 -> call -> 0x7ffffffffe448 # push on return address
rip = 0x555555555143 -> call -> 0x555555555139  # jump control to procedure
### AFTER CALL
return seven:
=> 0x5555555555139 <return_seven> mov
                                          $0x7, %eax
   0x55555555513e <return seven+5> retq
main: ...
   0x555555555513f <main>
                                   sub
                                          $0x8,%rsp
   0x55555555555143 < main+4>
                                   calla
                                          0x5555555555139 <return_seven>
   0x5555555555148 <main+9>
                                   lea
                                          0x2ee1(%rip),%rdi
   0x555555555514f <main+16>
                                          %eax.%esi
                                   mov
(gdb) x/gx $rsp
                                    # stack grew 8 bytes with call
0x7fffffffe448: 0x00005555555555148 # return address in main on stack
```

# return\_seven\_asm.s 2/2: Control Transfer with ret

```
### BEFORE RET
return seven:
                                         $0x7, %eax
   0x555555555139 <return seven> mov
=> 0x555555555513e <return seven+5> retq
main: ...
   0x555555555513f <main>
                                  sub
                                          $0x8, %rsp
   0x5555555555143 < main+4>
                                  calla
                                          0x5555555555139 <return seven>
   0x5555555555148 <main+9>
                                  lea
                                          0x2ee1(%rip),%rdi
                                  mov %eax,%esi
   0x555555555514f <main+16>
(gdb) x/gx $rsp
0x7fffffffe448: 0x0000555555555555148 # return address pointed to by %rsp
(gdb) stepi
                                                # EXECUTE RET INSTRUCTION
rsp = 0x7ffffffffe448 -> ret -> 0x7ffffffffe450
                                              # pops return address off
rip = 0x555555555513e -> ret -> 0x5555555555148
                                               # sets %rip to return address
### AFTER RET
return seven:
   0x555555555139 <return seven> mov $0x7, %eax
   0x55555555513e <return seven+5> reta
main: ...
   0x555555555513f <main>
                                          $0x8, %rsp
                                   sub
                                  callq
   0x55555555555143 < main+4>
                                          0x5555555555139 <return seven>
=> 0x5555555555148 <main+9>
                                  lea
                                          0x2ee1(%rip),%rdi
   0x55555555514f <main+16>
                                          %eax.%esi
                                  mov
(gdb) print $rsp --> $3 = 0x7fffffffe450
```

# Warning: %rsp is important for returns

- ► When a function is about to return %rsp MUST refer to the memory location of the return address
- ret uses value pointed to %rsp as the return address
- Segmentation Faults often occur if %rsp is NOT the return address - attempt to fetch/execute instructions out of bounds
- Using pushX / subq instructions to extend stack during a function MUST be coupled with popX / addq instructions
- ► There are computer security issues associated stack-based return value we will discuss later

# Messing up the Return Address

```
### return_seven_buggy_asm.s
.text
.global return_seven
return seven:
  pushq $0x42
                    # push but no pop before returning
  movl $7. %eax
                    # %rsp points to a 0x42 return address - BAD!
  ret.
 REG |
         VALUE |
                    I ADDRESS I
                                 VALUE | NOTE
 -----|
                    |-----|
| rax |
                    | 0x77128 | 0x554210 | Ret Address
| rsp | 0x77120 |--->| 0x77120 | 0x42 | Pushed Val
> gcc return_seven_buggy_asm.s
> ./a.out
Segmentation fault (core dumped) ## definitely a memory problem
> valgrind ./a.out
                                 ## get help from Valgrind
==2664132== Jump to the invalid address stated on the next line
==2664132==
              at 0x42: ???
                                 ## execute instruction at address 0x42??
==2664132== by 0x109149: ??? (in a.out)
==2664132== Address 0x42 is not stack'd, malloc'd or (recently) free'd
```

Valgrind output is not obvious but most reports like this indicate clobbering a return address as happened here through unbalanced push/pop instructions.

## Stack Alignment

- ► According to the strict x86-64 ABI, must align rsp (stack pointer) to 16-byte boundaries when calling functions
- Will often see arbitrary pushes or subtractions to align
  - Always enter a function with 8-byte Return Address on the stack
  - ▶ Means that it is aligned to 8-byte boundary
- rsp changes must be undone prior to return

- Failing to align the stack may work but may break
- ► Failing to "undo" stack pointer changes will likely result in return to the wrong spot : major problems

## x86-64 Register/Procedure Convention

- ▶ Used by Linux/Mac/BSD/General Unix
- ▶ Params and return in registers if possible

#### Parameters and Return

- First 6 arguments are put into
  - 1. rdi / edi / di (arg 1)
  - 2. rsi / esi / si (arg 2)
  - 3. rdx / edx / dx (arg 3)
  - 4. rcx / ecx / cx (arg 4)
  - 5. r8 / r8d / r8w (arg 5)
  - 6. r9 / r9d / r9w (arg 6)
- Additional arguments are pushed onto the stack
- ► Return Value in rax / eax /...

### Caller/Callee Save

Caller save registers: alter freely

rax rcx rdx rdi rsi r8 r9 r10 r11

**Callee** save registers: must restore these on return

rbx rbp r12 r13 r14 r15

Careful messing with stack pointer

rsp # stack pointer

## Pushing and Popping the Stack

- If local variables are needed on the stack, can use push / pop for these
- pushX %reg: grow rsp (lower value), move value to top of main memory stack,
  - pushq %rax: grows rsp by 8, puts contents of rax at top
  - pushl \$25: grows rsp by 4, puts constant 5 at top of stack
- popX %reg: move value from top of main memory stack to reg, shrink rsp (higher value)
  - popl %eax: move (%rsp) to eax, shrink rsp by 4

```
main:
    pushq
           %rbp
                             # save register, aligns stack
                             # like subg $8, %rsp; movg %rbp, (%rsp)
                             # call function
    call
            sum_range
            %eax, %ebp
    movl
                            # save answer
    . . .
    call
                             # call function, ebp not affected
            sum_range
            %rbp
    popq
                             # restore rbp, shrinks stack
                             # like movg (%rsp), %rbp; addg $8, %rsp
    ret
```

### Exercise: Local Variables which need an Address

### Compare code in files

- swap\_pointers.c : familiar C code for swap via pointers
- swap\_pointers\_asm.s : hand-coded assembly version

### Determine the following

- 1. Where are local C variables x,y stored in assembly version?
- 2. Where does the assembly version "grow" the stack?
- 3. How are the values in main() passed as arguments to swap\_ptr()?
- 4. Where does the assembly version "shrink" the stack?

### Exercise: Local Variables which need an Address

```
1 #include <stdio.h>
                                                  1 .text
2
                                                    .global swap ptr
   void swap_ptr(int *a, int *b){
                                                    swap_ptr:
     int tmp = *a;
                                                                     (%rdi), %eax
                                                  4
                                                             movl
                                                                     (%rsi), %edx
    *a = *b;
                                                             movl
    *b = tmp:
                                                             movl
                                                                     %edx. (%rdi)
     return;
                                                             movl
                                                                     %eax, (%rsi)
8 }
                                                             ret.
                                                    .global main
9 int main(int argc, char *argv[]){
10
     int x = 19;
                                                 10 main:
                                                                     $8, %rsp
11
    int y = 31;
                                                 11
                                                             suba
12
     swap_ptr(&x, &y);
                                                             movl
                                                                     $19, (%rsp)
                                                 12
13
     printf("%d %d\n",x,y);
                                                 13
                                                             movl
                                                                     $31, 4(%rsp)
14
     return 0:
                                                 14
                                                             movq
                                                                     %rsp, %rdi
15 }
                                                 15
                                                             leag
                                                                     4(%rsp), %rsi
                                                             call
                                                 16
                                                                     swap ptr
                                                 17
                                                 18
                                                             leag
                                                                     .FORMAT(%rip), %rdi
                                                                     (%rsp), %esi
                                                 19
                                                             movl
                                                                     4(%rsp), %edx
                                                 20
                                                             movl
                                                 21
                                                             movl
                                                                     $0, %eax
                                                 22
                                                             call
                                                                     printf@PLT
                                                 23
                                                                     $8, %rsp
                                                 24
                                                             addq
                                                                     $0. %eax
                                                 25
                                                             movl
                                                 26
                                                             ret
```

27 .data

.asciz "%d %d\n"

29

### **Answers**: Local Variables which need an Address

- 1. Where are local C variables x, y stored in assembly version?
- 2. Where does the assembly version "grow" the stack?
- 3. How are the values in main() passed as arguments to swap\_ptr()?

4. Where does the assembly version "shrink" the stack?

```
addq $8, %rsp  # shrink stack by 8 bytes
movl $0, %eax  # set return value
ret
```

## Diagram of Stack Variables

- Compiler determines if local variables go on stack
- ▶ If so, calculates location as rsp + offsets

```
1 // C Code: locals.c
                                        REG
                                               VALUE | Name
2 int set buf(char *b, int *s);
3 int main(){
                                             | #1024 | top of stack
                                        rsp
   // locals re-ordered on
                                                     | during main
5 // stack by compiler
6 int size = -1;
                                        MEM
7 char buf[16];
                                        #1031
                                                     | buf [3]
    int x = set_buf(buf, &size);
                                        #1030 I
                                                     | buf [2]
                                        #1029
                                                     | buf[1]
10 ## EQUIVALENT ASSEMBLY
                                        #1028
                                                     | buf [0]
14 main:
                             # space for buf/size and stack alignment
3
     subq $24, %rsp
     movl $-1,(%rsp)
                             # old rip already in stack so: 20+4+8 = 32
                             # initialize buf and size: main line 6
      . . . .
     leag 4(%rsp), %rdi # address of buf arg1
6
            0(%rsp), %rsi # address of size arg2
     leag
     call set buf
                            # call function, aligned to 16-byte boundary
     movl %eax, %r8
                             # get return value
10
     addq
             $24, %rsp # shrink stack size
11
```

# Summary of Procedure Calls: ABC() calls XYZ()

ABC() Caller callq XYZ # ABC to XYZ
XYZ() Callee retq # XYZ to ABC

- ABC() "saves" any Caller Save registers it needs by either copying them into Callee Save registers or pushing them into the stack
- ABC() places up to 6 arguments in %rsi, %rdi, %rdx, ..., remaining arguments in stack
- ABC() ensures that stack is "aligned": %rsp contains an address that is evenly divisible by 16
- 4. ABC() issues the callq ABC instruction which (1) grows the stack by subtracting 8 from %rsp and copies a return address to that location and (2) changes %rip to the staring address of func
- 5. XYZ() now has control: %rip points to first instruction of XYZ()
- XYZ() may issue pushX val instructions or subq N,%rsp instructions to grow the stack for local variables
- XYZ() may freely change Caller Save registers BUT Callee Save registers it changes must be restored prior to returning.
- XYZ() must shrink the stack to its original position via popX %reg or addq N,%rsp instructions before returning.
- 9. XYZ() sets %rax / %eax / %ax to its return value if any.
- 10. XYZ() finishes, issues the retq instruction which (1) sets the %rip to the 8-byte return address at the top of the stack (pointed to by %rsp) and (2) shrinks the stack by doing addq \$8,%rsp
- ABC() function now has control back with %rip pointing to instruction after call XYZ; may have a return value in %rax register
- 12. ABC() must assume all Caller Save registers have changed

# Historical Aside: Base Pointer rbp was Important

```
int bar(int, int, int);
int foo(void) {
  int x = callee(1, 2, 3);
  return x+5;
}
```

- 32-bit x86 / IA32 assembly used rbp as bottom of stack frame, rsp as top.
- Push all arguments onto the stack when calling changing both rsp and rbp
- x86-64: default rbp to general purpose register, not used for stack tracking

```
# Old x86 / IA32 calling sequence: set both %esp and %ebp for function call
foo:
   pushl %ebp
               # modifying ebp, save it
   ## Set up for function call to bar()
   movl %esp,%ebp
                       # new frame for next function
   pushl 3
                          # push all arguments to
    pushl 2
                          # function onto stack
   pushl 1
                          # no regs used
    call bar
                          # call function, return val in %eax
    ## Tear down for function call bar()
                          # restore stack top: args popped
   movl %ebp,%esp
    ## Continue with function foo()
   addl 5.%eax # add onto answer
   popl %ebp
                          # restore previous base pointer
   ret.
```